HP-41CV/CX
Triangle Solutions

```
9+00.000    100.25
9+25.000    107.88
9+50.000    107.50
9+75.000    107.13
10+00.000   106.75
10+25.000   106.38

10+50.000 BVC  106.00
 200 FT V.C.
11+50.000 PVI  104.50
          POVC 105.88
11+04.545 G=0% 105.59
```

prompt: STA?

keystrokes:

2 5 I

output:

```
10+75.000   105.71
11+00.000   105.59
11+25.000   105.65
```



Software by D'Zign
P.O. BOX 1370 • PACIFICA, CA 94044

NOTICE

No express or implied warranty is made by D'Zign Land Survey & Development or the author with regard to the procedures and program material offered or their merchantability or their fitness for any particular purpose. The keystroke procedures and program material are made available solely on an "as-is" basis, and the entire risk as to their quality and performance is with the user. Should the procedures or program material prove defective, the user (and not D'Zign Land Survey & Development nor any other party) shall bear any and all cost of all necessary correction and all incidental or consequential damages. D'Zign Land Survey & Development and/or the author shall not be liable for any incidental or consequential damages in connection with or arising out of the furnishing, use, or performance of the keystroke procedures or program material.

HP-41CV/CX
Triangle Solutions

published by D'Zign **Land Survey & Development**
Pacifica, California 1987

ISBN 0-9616846-9-0

The program for triangle solutions in this booklet has been through a number of changes over the years. The original version was written for the HP-65, when the Hewlett-packard version required the angle input to be in decimal degrees. An updated version was included in the book "Surveying Field Solutions", at which time we published the AREA, SIDE, SIDE and AREA, SIDE, ANGLE solutions for the first time.

Another solution has been added, AREA, ANGLE, ANGLE, so that any triangle with ANY three knowns may be solved. The version contained in this booklet is similar to the one in our surveying module, COGO 41.

The notations for angles and sides is familiar to HP users, but is not the standard, or 'textbook', notation which you have learned in trigonometry (side **a** opposite angle **A**, side **b** opposite angle **B** and side **c** opposite angle **C**). The sides and angles are numbered, in order, going around the triangle.



The example triangle (above) shows this style of labeling, compared to the standard notation for sides and angles. **Side 1** may be assigned to any side that is convenient to use, depending upon the available information about the triangle. It should be located at a side where the known information then falls into position for solution by one of the routines.



In the example, the assigned designations go clockwise. If it will better fit the information available, it may go counter-clockwise instead, as shown to the left.

Introduction

The **USE** of this program is simplicity itself. You need three known parts of the triangle being solved, and there are only three keys used for input.

If the input is an angle, stroke [A], if the input is a side, you stroke [B], and if the input is an area, you stroke [C]. The only other keys used are [E] (for re-start) and, if no printer is attached, [R/S].

These keystrokes are shown in the sketch to the right, and keystroke examples have been included with the descriptions of each type of solution.



The program may be accessed and initialized by stroking [XEQ] [ALPHA] [T] [R] [ALPHA]. Before starting, the calculator should be sized to at least a size of **010**. Additional registers may be used for scratch by sizing at a higher number. The registers used by the program are **00** through **09**.

Once the program has been accessed, stroking [E] after completing a solution will clear the calculator for the next solution. If the program is used without a printer attached, stroke [R/S] as the output is displayed.

The triangle shown to the right will be used for the examples. It should be noted that the output will vary slightly, depending on the number of places input, particularly in the input of the angles.



Side 1, Side 2, Side 3

THREE SIDES KNOWN is one of the most used solutions for triangles, particularly in recent years in surveying.

The lower cost and higher accuracy of electronic distance measurement equipment has resulted in more trilateration being used, instead of time-consuming repetitions of the angles.

keystroke example:



output:

```
S1=83.6400
A1=36°12'30.0"

S2=96.8000
A2=59°19'12.1"

S3=57.4492
A3=84°28'17.9"

AREA=2,391.3506
```



This routine uses the equations

$$A_3 = 2\cos^{-1}\sqrt{\frac{P(P - S_2)}{S_1 S_3}}$$

$$A_2 = 2\cos^{-1}\sqrt{\frac{P(P - S_1)}{S_2 S_3}}$$

and

$$A_1 = \cos^{-1}(-\cos(A_3 + A_2))$$

where $P = \frac{1}{2}(S_1 + S_2 + S_3)$

Side 1, Angle 1, Side 2

TWO SIDES AND THE INCLUDED ANGLE KNOWN is resolved by finding the third side, and then solving the triangle as shown on the previous page. The third side is found through the use of the equation

$$S_3 = \sqrt{S_1^2 + S_2^2 - 2\,S_1 S_2 \cos A_1}$$



keystrokes:



output:

```
S1=57.4492
A1=84°28'18.0"

S2=83.6400
A2=36°12'30.0"

S3=96.8000
A3=59°19'12.0"

AREA=2,391.3506
```

NOTE: areas are calculated by this program using the equation shown below.



$$\text{Area} = \tfrac{1}{2}(S_1\, S_3 \sin A_3)$$

Side 1, Angle 1, Angle 2



ONE SIDE AND THE TWO FOLLOWING ANGLES KNOWN. This solution first solves for the third angle with the equation

$$A_3 = \cos^{-1}\left(-\cos\left(A_1 + A_2\right)\right)$$

Once angle 3 has been found, the remainder of the triangle is solved as Angle, Side, Angle (see page 8 for equations) to determine the other missing sides.

keystrokes:

Stroke XEQ ALPHA T R ALPHA to bring the program to the top of the program memory or, if you have already been using this program, stroke E

5 7 · 4 4 9 2
B

8 4 · 2 8 1 8
A

3 6 · 1 2 3
A

to reset for another solution, stroke E

output:

```
S1=57.4492
A1=84°28'18.0"

S2=83.6400
A2=36°12'30.0"

S3=96.8000
A3=59°19'12.0"

AREA=2,391.3500
```

Side 1, Side 2, Angle 2

TWO SIDES AND THE FOLLOWING ANGLE KNOWN has two possible solutions. When this configuration is used, both solutions are output. The second solution will not necessarily show the parts in the same order as the input.

The other two angles are calculated with the equations below, and the remaining side is calculated as an Angle, Side, Angle configuration.

$$A_3 = \sin^{-1}\left[\frac{S_2}{S_1}\sin A_2\right] \qquad A_1 = \cos^{-1}\left[-\cos\left(A_2 + A_3\right)\right]$$



output:

```
S1=57.4492
A1=84°28'17.9"

S2=83.6400
A2=36°12'30.0"

S3=96.8000
A3=59°19'12.1"

AREA=2,391.3505

2ND SOLUTION:
S1=83.6400
A1=36°12'30.0"

S2=38.1740
A2=120°40'47.9"

S3=57.4492
A3=23°06'42.1"

AREA=943.0510
```

keystrokes:

Stroke XEQ ALPHA T R ALPHA, bring the program to the top of the program memory or, if you have already been using this program, stroke E

5 7 · 4 4 9 2 B

8 3 · 6 4 B

3 6 · 1 2 3 A

Angle 3, Side 1, Angle 1

TWO ANGLES AND THE INCLUDED SIDE ARE KNOWN.

$$S_2 = S_1 \frac{\sin A_3}{\sin A_2}$$

$$S_3 = S_1 \cos A_3 + S_2 \cos A_2$$

$$A_2 = \cos^{-1}(-\cos(A_3 + A_1))$$

This configuration is solved by using the equations shown to the left.

The Angle, Side, Angle routine has also been used as a secondary solution to some of the other routines, after the problem has first been reduced to these three known parts.

keystrokes:

5 9 . 1 9 1 2 A

5 7 . 4 4 9 2 B

8 4 . 2 8 1 8 A

output:

```
S1=57.4492
A1=84°28'18.0"

S2=83.6400
A2=36°12'30.0"

S3=96.8000
A3=59°19'12.0"

AREA=2,391.3500
```



Area, Side 1, Angle 1

THE AREA, ONE SIDE AND ONE ANGLE KNOWN is the first of the three routines in this program which allow the area to be used as one of the known parts. Whenever the area <u>is</u> one of the parts, it is input first.

The equation

$$S_2 = \frac{2\ AREA}{S_1 \sin A_1}$$

is used first to reduce the problem for solution as Side 1, Angle 1, Side 2.



keystrokes:

Stroke XEQ ALPHA T R ALPHA to bring the program to the top of the program memory or, if you have already been using this program, stroke E

2 3 9 1 . 3 5 C

9 6 . 8 B

5 9 . 1 9 1 2 A

output:

```
S1=96.8000
A1=59°19'12.0"

S2=57.4492
A2=84°28'18.0"

S3=83.6400
A3=36°12'30.0"

AREA=2,391.3500
```

Area, Angle 3, Angle 1

AREA AND TWO ANGLES KNOWN is first solved for the included side, and then solved as Angle, Side, Angle. The first angle input is treated as Angle 3, the second as Angle 1. The equation used for finding side 1 is

$$S_1 = \sqrt{\frac{2\sin A_2(AREA)}{\sin A_1 \sin A_3}} \quad \text{where } A_2 = \cos^{-1}(-\cos(A_1 + A_3))$$



keystrokes:

Stroke XEQ ALPHA T R ALPHA or E to reset the flag status, if you have already been using the program, and then begin with the input of the area.

2 3 9 1 . 3 5 C
5 9 . 1 9 1 2 A
8 4 . 2 8 1 8 A

output:

```
S1=57.4492
A1=84°28'18.0"

S2=83.6400
A2=36°12'30.0"

S3=96.8000
A3=59°19'12.0"

AREA=2,391.3500
```

Area, Side 1, Side 2

AREA AND TWO SIDES KNOWN is another problem which has two possible solutions.

We first find Angle 1 with the equation

$A_1 = \sin^{-1}\left[\frac{2AREA}{S_1\ S_2}\right]$ and then solve as Side, Angle, Side. The second solution is possible where angle 1 may also be equal to 180° - angle 1. This value is substituted, and the second solution is output.



keystrokes:

E or XEQ ALPHA T R ALPHA
2 3 9 1 . 3 5
C
9 6 . 8
B
5 7 . 4 4 9 2
B

output:

```
S1=96.8000
A1=59°19'12.0"

S2=57.4492
A2=84°28'18.0"

S3=83.6400
A3=36°12'30.0"

AREA=2,391.3500

2ND SOLUTION:
S1=96.8000
A1=120°40'48.0"

S2=57.4492
A2=37°55'33.4"

S3=135.4461
A3=21°23'38.6"

AREA=2,391.3500
```

Program Listings

The listing below contains the program steps for the triangle program, "**TR**". The program, together with its subroutine, "**DMS**" contains 400 program steps (626 bytes), and occupies 89 registers.

If your version of this booklet includes the programmed magnetic cards, simply input the program from them. If you have the cards, they include the synthetic steps which output the angles as 36°12'32.0". The listing below will output the angles in the form 36 12 32.0 as written.

If you can input the synthetics, thay may be substituted as follows: change step 379 to hex F2 7F 7A (or any other symbol you like, the 41 doesn't contain a degree symbol), change step 387 to hex F2 7F 27, and insert hex F2 7F 22 between steps 389 and 390.

There are no 'special instructions' for input of the steps, and all of the functions used are standard on any version of the HP 41.

```
01♦LBL "TR"     21 SF 03        41 GTO 10       61 GTO 18
02 .002         22 FS? 01       42♦LBL C        62 FS? 02
03 STO 00       23 SF 02        43 SF 07        63 GTO 17
04 CLX          24 SF 01        44 ISG 00       64 GTO 16
05 CF 00        25 ISG 00       45 RTN          65 RTN
06 CF 01        26 RTN          46♦LBL 10       66♦LBL 07
07 CF 02        27 GTO 10       47 FS? 07       67 FS? 02
08 CF 03        28 RTN          48 GTO 06       68 GTO 14
09 CF 04        29♦LBL A        49 FS? 09       69 FS? 03
10 CF 05        30 FS? 02       50 GTO 11       70 GTO 12
11 CF 06        31 SF 08        51 FS? 08       71 GTO 13
12 CF 07        32 FS? 01       52 GTO 15       72 RTN
13 CF 08        33 SF 06        53 FS? 06       73♦LBL 11
14 CF 09        34 FS? 02       54 GTO 07       74 STO 04
15 SF 21        35 CF 06        55 FS? 03       75 RDN
16 SF 27        36 FS? 04       56 GTO 12       76 STO 02
17 FIX 4        37 SF 05        57 GTO 11       77 RDN
18 RTN          38 SF 04        58 RTN          78 STO 09
19♦LBL B        39 ISG 00       59♦LBL 06       79 RDN
20 FS? 04       40 RTN          60 FS? 05       80 RDN
```

```
 81 +
 82 +
 83 2
 84 /
 85 STO 07
 86 X↑2
 87 LASTX
 88 RCL 02
 89 *
 90 -
 91 RCL 09
 92 RCL 04
 93 *
 94 /
 95 SQRT
 96 ACOS
 97 2
 98 *
 99 STO 05
100 SIN
101 RCL 09
102 *
103 STO 08
104 RCL 07
105 X↑2
106 LASTX
107 RCL 09
108 *
109 -
110 RCL 02
111 /
112 RCL 04
113 /
114 SQRT
115 ACOS
116 2
117 *
118 STO 03
119 RCL 05
120 XEQ 00
121 STO 01
122 GTO 01
123 RTN
124♦LBL 12
125 "2ND SOLUTION:"
126 FS?C 00
127 AVIEW
128 HR
129 STO 01
130 RDN
131 STO 09
132 RDN
133 HR
134 STO 05
135♦LBL 03
136 RCL 01
137 XEQ 00
138 STO 03
139 RCL 05
140 RCL 09
141 P-R
142 X<>Y
143 STO 08
144 RCL 03
145 1
146 P-R
147 RDN
148 /
149 STO 02
150 R↑
151 *
152 +
153 STO 04
154 GTO 01
155 RTN
156♦LBL 13
157 HR
158 STO 03
159 RDN
160 HR
161 STO 01
162 RDN
163 STO 09
164 RCL 03
165 RCL 01
166 XEQ 00
167 RCL 09
168 RCL 01
169 XEQ 04
170 GTO 03
171 RTN
172♦LBL 14
173 STO 02
174 RDN
175 HR
176 STO 01
177 RDN
178 STO 09
179 RCL 01
180 RCL 02
181 P-R
182 RCL 09
183 -
184 R-P
185 STO 04
186 RCL 09
187 RCL 02
188 RCL 04
189 GTO 11
190 RTN
191♦LBL 15
192 HR
193 STO 03
194 RDN
195 STO 02
196 RDN
197 STO 09
198 RCL 03
199 SIN
200 RCL 02
201 *
202 RCL 09
203 /
204 ASIN
205 STO 05
206 RCL 03
207 XEQ 00
208 STO 01
209 RCL 05
210 RCL 09
211 RCL 01
212 XEQ 04
213 XEQ 03
214 180
215 RCL 05
216 -
217 RCL 03
218 +
219 CHS
220 180
221 +
222 HMS
223 RCL 02
224 RCL 03
225 HMS
226 SF 00
227 GTO 12
228 RTN
229♦LBL A
230 GTO A
231♦LBL B
232 GTO B
233♦LBL 00
234 +
235 COS
236 CHS
237 ACOS
238 RTN
239♦LBL 01
240 "S1="
```

Program Listings

```
241 ARCL 09
242 AVIEW
243 RCL 01
244 HMS
245 CLA
246 "A1="
247 XEQ "DMS"
248 ADV
249 "S2="
250 ARCL 02
251 AVIEW
252 RCL 03
253 HMS
254 CLA
255 "A2="
256 XEQ "DMS"
257 ADV
258 "S3="
259 ARCL 04
260 AVIEW
261 RCL 05
262 HMS
263 CLA
264 "A3="
265 XEQ "DMS"
266 ADV
267 RCL 08
268 RCL 04
269 *
270 2
271 /
272 "AREA="
273 ARCL X
274 AVIEW
275 ADV
276 FS? 10
277 XEQ 08
278 RTN
279♦LBL 08
280 "2ND SOLUTION:"
281 FS?C 10
282 AVIEW
283 180
284 RCL 01
285 -
286 RCL 09
287 X<>Y
288 HMS
289 RCL 02
290 GTO 14
291 RTN
292♦LBL 09
293 RDN
294 RDN
295 RTN
296♦LBL 04
297 STO 01
298 RDN
299 STO 09
300 RDN
301 STO 05
302 RTN
303♦LBL 16
304 HR
305 STO 01
306 SIN
307 XEQ 02
308 STO 02
309 XEQ 05
310 RTN
311♦LBL 17
312 SF 10
313 STO 02
314 XEQ 02
315 ASIN
316 STO 01
317 XEQ 05
318 RTN
319♦LBL 18
320 HR
321 STO 03
322 X<>Y
323 HR
324 STO 01
325 +
326 180
327 X<>Y
328 -
329 STO 05
330 SIN
331 2
332 *
333 *
334 RCL 01
335 SIN
336 RCL 03
337 SIN
338 *
339 /
340 SQRT
341 STO 02
342 RCL 01
343 HMS
344 X<>Y
345 RCL 03
346 HMS
347 XEQ 12
348 RTN
349♦LBL 02
350 X<>Y
351 STO 09
352 *
353 /
354 2
355 *
356 RTN
357♦LBL 05
358 RCL 09
359 RCL 01
360 HMS
361 RCL 02
362 XEQ 14
363 RTN
364♦LBL E
365 XEQ "TR"
366 RTN
367♦LBL "DMS"
368 ENTER↑
369 INT
370 CF 29
371 FIX 0
372 ARCL X
373 -
374 1 E2
375 *
376 ABS
377 STO 06
378 INT
379 "├ "
380 XEQ 01
381 ARCL X
382 RCL 06
383 FRC
384 1 E2
385 *
386 FIX 1
387 "├ "
388 XEQ 01
389 ARCL X
390 AVIEW
391 FIX 4
392 SF 29
393 RTN
394♦LBL 01
395 10
396 X<>Y
397 X<Y?
398 "├0"
399 RTN
400 END
```

HP41-CV/CX
SOLUTIONS BOOKLETS

VERTICAL ALIGNMENT
Calculates CONTINUOUS vertical alignment without changing back and forth between Grade and Curve routines. Asymmetrical, Compound and Reverse Curves.

booklet #525 . . . . . **$6.95**
with programmed cards **$12.95**

SPIRAL CURVE SOLUTIONS
The only COMPLETE spiral curve program for use with hand-helds.

Booklet #529 . . . . . **$6.95**
with programmed cards **$12.95**

EDM SLOPE STAKING
Sets slope stakes and reference points from remote instrument and backsight locations directly.

Booklet #533 . . . . . **$6.95**
with programmed cards **$12.95**

TRIANGLE SOLUTIONS
The most complete triangle solutions program available. Solves with any of the following knowns: ASA SAA SAS SSA SSS Area-SS Area-AA Area-SA

Booklet #535 . . . . . **$6.95**
with programmed cards **$12.95**

POINT STORAGE CONVERSIONS
for the
HP41-CV/CX Surveying Pac

Traverse, Inverse & Sideshots
w/automatic compass correction.

Booklet #521 . . . . . **$6.95**
with programmed cards **$12.95**

INTERSECTION SOLUTIONS
(Auto-Inverse routine added)
Bearing-Bearing, Bearing-Dist, Dist-Dist and Offset to a Line.

Booklet #523 . . . . . **$6.95**
with programmed cards **$12.95**

Coordinate Transformation
Rotate or re-scale. Automatically outputs the new rotated bearings and distances between points.

Booklet #527 . . . . . **$6.95**
with programmed cards **$12.95**

PREDETERMINED AREAS
Calculate required areas from your pre-stored coordinate pairs.

Booklet #531 . . . . . **$6.95**
with programmed cards **$12.95**

Software by D'Zign
P.O. BOX 1370 • PACIFICA, CA 94044
[415] 355-8942

CA residents add 6.5% sales tax

COCO 41

This is the ROM you've been needing. Complete traverse, inverse . . . .everything the others have and more! Coordinate storage and retrieval option by **point number**. No need to pre-divide to mean angles before input in the field.

The closure routine includes **automatic angle check and adjustment**. The compass and transit corrections are automatic, with **auto-inverse** between the adjusted coordinates, all at the touch of a button. Does not require X-memory to work.

Contains keyboard functions for all of the common conversions; Feet to Meters, Feet to Foot-Inch-Fraction, Azimuth to Bearing, (option of North or South Azimuth), add or subtract feet-inch-fractions directly with one shifted keystroke! And, of course, inverse functions of all of these.

available early 1988

ISBN 0-9616846-9-0